This example demonstrates a generic way to enable information ‘sharing’ by e-mail or message(SMS). 
Imagine an in-field staff member using your iPhone or iPad App wants to: 
· ‘Share’ the details of an overdue Order with your billing department by e-mail 
· ‘Share’ the details of a customer with another sales rep by SMS   
· ‘Share’ the details of a spare part enquiry with a potential customer by SMS 

Here is a screen that mocks up all 3 of those situations ........

When the user touches the Spare Part’s ‘Share’ button a pop over appears asking whether they want to share the details by by e-mail or message (sms) ..........
[image: ]
When they touch “Share by E-mail” an e-mail message prefilled with the share details is presented read to send .........           
[image: ]
In this case what is ‘shared’ in the e-mail is very similar to what was on the screen – but you can of course share more or less details than were visible on the screen.     
The following code is useful because it is generic – you could use some thing like it to add a ‘sharing’ capability to your existing Form Views in a very short time. 
EXAM0075A (RPG) – Generic Subroutine - Sets up share button(s) on any screen 

     H BNDDIR('LRNGSRV') DFTACTGRP(*no) ACTGRP(*caller)
      * Data structures, variables, constants and prototypes
     D/INCLUDE LRNGSOURCE,LRNGCOMMON
     D/INCLUDE SOURCEDEMO,DEMOCOMMON
     C* Parameter list declaration
     C     *Entry        PLIST
     C                   Parm                    Container       256
     C                   Parm                    Title           256
     C                   Parm                    Message         256
      /free

        /// This generic program inserts a share button into the
        /// caller's form view. The caller provides the fully qualified
        /// <<Name>> of the container on its screen that will have the
        /// button inserted. The caller also provides the message that
        /// should be shared if the user touches the button

        // Set up the associated container with a button

        LRNG_Using(%trim(Container));
          LRNG_SETProperty('Type'            : 'Button');
          LRNG_SetProperty('OnClick'         : 'EXAM0075B_SHOW_SHARE_METHODS');
          LRNG_SetProperty('Text'            : 'Share' ) ;
          LRNG_SetProperty('OnClick.Param%1' : %trim(Container));
          LRNG_SetProperty('OnClick.Param%2' : %trim(Title));
          LRNG_SetProperty('OnClick.Param%3' : %trim(Message));
        LRNG_EndUsing();

        // Finished

        *InLR = False;
        Return;

      /end-free
EXAM0075B (RPG) – Generic Subroutine - Handle the user touching a share button on any screen

     H BNDDIR('LRNGSRV') DFTACTGRP(*no) ACTGRP(*caller)
      * Data structures, variables and prototypes
     D/INCLUDE LRNGSOURCE,LRNGCOMMON
     D PNL             C                   X'25'
     D                 DS
     D Container                    256A
     D Title                        256A
     D Message                      256A
     D/INCLUDE SOURCEDEMO,DEMOCOMMON
     C* Parameter list declaration
     C     *Entry        PLIST
     C                   Parm                    RequestACTION    32
      /free

        /// This generic program acts in concert with EXAM0075A. It handles the
        /// user touching the button 'Share' button to share details.

        if (RequestACTION = 'EXAM0075B_SHOW_SHARE_METHODS');
           Exsr SHOW_SHARE_METHODS;
        endif;

        // Finished

        *InLR = False;
        Return;

        /// ----------------------------------------------------------
        /// Create the form required to select E-Mail or SMS
        /// ----------------------------------------------------------
        //
        Begsr SHOW_SHARE_METHODS;
         // Get the associated event parameter details that would
         // have been set up by EXAM0075A. These tell this program
         // the visual container that initiated the request and
         // the title and message of what is to be shared

         LRNG_Using('/CurrentOp');
         LRNG_GetProperty('Param%1' : Container : %size(Container));
         LRNG_GetProperty('Param%2' : Title     : %size(Title));
         LRNG_GetProperty('Param%3' : Message   : %size(Message));
         LRNG_EndUsing();

         // Build the pop over form with the e-mail and SMS button options

         LRNG_Using(%trim(Container));

         LRNG_SETProperty('ShowPopOver':'True');

         LRNG_Using('PopOver');

              // The pop over is a form by itself
              LRNG_SetProperty('Type'              : 'Form');
              LRNG_SetProperty('BackColor'         : '&0d1a35');

              // Add the e-mail button
              LRNG_Using('Fields.EMAIL');
                LRNG_SetProperty('Type'        : 'Button');
                LRNG_SetProperty('Text'        : 'Share by E-mail');
                LRNG_SetProperty('Layout.Row'  : '1');
                LRNG_Using('OnClick.Operation');
                   LRNG_SetProperty('Type'     : 'SendMail');
                   LRNG_SetProperty('Subject'  : Title);
                   LRNG_SetProperty('Body'     : Message);
                LRNG_EndUsing();
              LRNG_EndUsing();

              // Add the sms button
              LRNG_Using('Fields.SMS');
                LRNG_SetProperty('Type'        : 'Button');
                LRNG_SetProperty('Text'        : 'Share by Message');
                LRNG_SetProperty('Layout.Row'  : '2');
                LRNG_Using('OnClick.Operation');
                   LRNG_SetProperty('Type'     : 'SendTextMessage');
                   LRNG_SetProperty('Message' : %trimr(Title)+PNL+Message);
                LRNG_EndUsing();
              LRNG_EndUsing();

         LRNG_EndUsing();

        Endsr;

      /end-free
EXAM0075D (DSPF) and EXAM0075 (RPG) – Mock up Form View to demonstrate use of the generic subroutines 

     A                                       DSPSIZ(27 132)
     A* -----------------------------------------------------------------------
     A          R FORMVIEW
     A* Put the details of a dummy customer, order, invoice and part
     A* on the screen. beside each create a <<tag>> name that EXAM0075A
     A* will dynamically convert to a 'Share'button
     A                                  2 10'<<TITLE>>'
     A                                  2 10HTML('Type:Label Color:Blue
     A                                      Font.size:18 Layout.Cols:-1 ')
     A* A dummy example customer
     A                                  4 10HTML('Type:Label BackColor:&b8bbbf
     A                                      Layout.Height:1 Layout.Cols:-1')
     A                                  5 10'Customer'
     A                                  5 10HTML('Color:Blue Font.Size:20')
     A            CUSTNUM       10   O  5 40
     A                                  5 60'<<CUSTOMER_SHARE>>'
     A            CUST01        50      6 10
     A            CUST02        50      7 10
     A            CUST03        50      8 10
     A* A dummy example order number
     A                                  9 10HTML('Type:Label BackColor:&b8bbbf
     A                                      Layout.Height:1 Layout.Cols:-1')
     A                                 10 10'Order'
     A                                 10 10HTML('Color:Blue Font.Size:20')
     A            ORDERNUM      10   O 10 40
     A                                 10 60'<<ORDER_SHARE>>'
     A            ORDR01        50     11 10
     A            ORDR02        50     12 10
     A            ORDR03        50     13 10
     A* A dummy example spare part number
     A                                 14 10HTML('Type:Label BackColor:&b8bbbf
     A                                      Layout.Height:1 Layout.Cols:-1')
     A                                 15 10'Spare Part'
     A                                 15 10HTML('Color:Blue Font.Size:20')
     A            PARTNUM       10   O 15 40
     A                                 15 60'<<PART_SHARE>>'
     A            PART01        50     16 10
     A            PART02        50     17 10
     A            PART03        50     18 10
     A                                 19 10HTML('Type:Label BackColor:&b8bbbf
     A                                      Layout.Height:1 Layout.Cols:-1')



     H BNDDIR('LRNGSRV') DFTACTGRP(*no) ACTGRP(*caller)
      * RPG file declarations - includes all form view definition(s)
     FEXAM0075D CF   E             WORKSTN
      * Standard include to bring in the external LRNG support functions
      /INCLUDE LRNGSOURCE,LRNGCOMMON
      * Standard demonstration definitions and prototypes
      /INCLUDE SOURCEDEMO,DEMOCOMMON
      * Local Variables
     D PLF             C                   X'25'
      * Standard form view handler parameter list declaration
      /INCLUDE SOURCEDEMO,STDPARAMS
      * Use free format coding style
      /free

         // Set up dummy customer, order, invoice and part numbers to act as
         // examples of the type of data that might be displayed by this program
         // ready for sharing via e-mail or SMS

         CustNum    = '1234567';
           Cust01   = 'ACME Engineering';
           Cust02   = '121 Some Street';
           Cust03   = 'Somewhere';

         OrderNum   = '61-5363';
           Ordr01   = 'ADT-Finch Consulting';
           Ordr02   = 'Placed:20th May, Due:25th May, Shipped:YES';
           Ordr03   = 'Value:56,536.56, Tax:647.56, CLim:120';

         PartNum    = '633Q63A';
           Part01   = 'Green Widget Spangler';
           Part02   = 'In Stock : Yes, Var Sizes : True, Rets : No';
           Part03   = 'http://www.mysite.com/products?id=633Q63A';

         // Loop for as long as this program is the one handling the requests
         DoU (RequestPROGRAM <> ProgramINFO.Name);
            // Handle the requested action
            Exsr HandleRequestedAction;
            // Display the resulting form view
            Exsr DisplayFormView;
         Enddo;
         // Return control back to the driver
         *InLR = True;
         Return;

       // ======================================================================
       // HandleRequestedAction : Handle the user's requested action
       // ======================================================================

       Begsr HandleRequestedAction;

         // Initially assume that the information displayed will be changed
         IsNewForm = True;

         // Now handle the user's requested action (found in RequestACTION)
         Select;

           // If the user has requested to see RPG or DDS Code
           // NOTE: This code would not be in a real application of course!
           When ( %Subst(RequestACTION:1:4) = 'SHOW');
               DemoShowMember( SystemInfo :  RequestACTION
                             : ProgramINFO.Name : RequestWasHandled );
               IsNewForm = False;

           /// If the request starts with EXAM0075B delegate it
           When ( %Subst(RequestACTION:1:9) = 'EXAM0075B');
              EXAM0075B(RequestACTION);
              IsNewForm = False;


           // Otherwise set up the 3 different share buttons on the screen
           Other;
              // Look at the DDS in member EXAM0075D to see these buttons
              // The parameters are the name of the button on this screen,
              // the title of what is to be shared and the message (content)
              // of what is to be shared.

              EXAM0075A ('/Form.Fields.ORDER_SHARE'
                       : 'Your Order : ' + %trimr(OrderNum)
                       : %trimr(Ordr01)+PLF+%trimr(Ordr02)+PLF+%trimr(Ordr03));

              EXAM0075A ('/Form.Fields.CUSTOMER_SHARE'
                       : 'Details of customer : ' + %trimr(CustNum)
                       : %trimr(Cust01)+PLF+%trimr(Cust02)+PLF+%trimr(Cust03));

              EXAM0075A ('/Form.Fields.PART_SHARE'
                   : 'Part enquiry for ' + %trimr(PartNum)
                   : %trimr(Part01)+PLF+%trimr(Part02)+PLF+PLF+%trimr(Part03));

              // End of selection
          EndSl;

           // Set up the title

          LRNG_SetProperty('/Form.Fields.TITLE.Value'
                : 'Touch the share button to share the details'
                + ' of the displayed customer, order or spare part');

       Endsr;

       // ----------------------------------------------------------------------
       // Display the formview
       // ----------------------------------------------------------------------
       Begsr DisplayFormView;

          // Write out and read back this handler's form view (screen layout).
          If (Not IsNewForm);
             LRNG_SetSendChangesOnly();
          Endif;
          LRNG_Send();
          // Typically only the DDS record format name is changed or SELECTed.
          EXFMT FORMVIEW;
          LRNG_Receive();
          // Get the next requested action to take and the program to handle it
          LRNG_GetRequestedAction(RequestPROGRAM:RequestACTION);
       Endsr;


      /end-free

